



# Smart Contract Security Audit Report

**Prepared for Litra Finance**

**Prepared by Supremacy**

March 23, 2025

## Contents

1 Introduction .....	3
1.1 About Client .....	4
1.2 Audit Scope .....	4
1.3 Changelogs .....	4
1.4 Threat Model .....	5
1.5 About Us .....	5
1.6 Terminology .....	5
2 Findings .....	7
2.1 Critical .....	8
2.2 Medium .....	10
2.3 Low .....	13
2.4 Informational .....	16
3 Disclaimer .....	19

# **1 Introduction**

Given the opportunity to review the design document and related codebase of the Litra Finance, we outline in the report our systematic approach to evaluate potential security issues in the smart contract(s) implementation, and provide additional suggestions or recommendations for improvement. Our results show that the given version of smart contracts can be further improved due to the presence of several issues related to either security or performance. This document outlines our audit results.

## 1.1 About Client

Litra Finance is an NFT liquidity protocol that wraps NFT into fungible ERC20 tokens to improve trading accuracy and provide more basic liquidity targets. Combining an automated centralized liquidity AMM with a customized curve provides traders with the ultimate trading experience with low slippage, while reducing impermanent losses for liquidity providers. The introduction of the veToken model encourages liquidity providers to actively and consistently provide liquidity for greater returns, which also aligns the interests of all parties.

Item	Description
Client	Litra Finance
Type	Smart Contract
Languages	Solidity
Platform	EVM-compatible

## 1.2 Audit Scope


In the following, we show the given compressed file with codebase and its checksum value before and after auditing.

- litra-contracts.rar (SHA256: 0a3e607cfff4605d061e...fb4019b0dc4ce1b929d8)
- litra-contracts.rar (SHA256: 9303970128295bc8c223...1798eca9a38f97643b5b)

## 1.3 Changelogs

Version	Date	Description
0.1	April 07, 2023	Initial Draft
0.2	April 08, 2023	Release Candidate #1
1.0	April 23, 2023	Final Release
1.1	March 23, 2025	Post-Final Release #1

## 1.4 Threat Model



Litra Finance is an NFT liquidity protocol, and within the scope of observable security audits its main functions are the components NFTVault, FeeManager and Admin.

As shown above, this involves multiple interactions between a user who (wraps) his NFT into a WrappedNFT via Litra Finance and a user who (unwraps) his WrappedNFT into an NFT via Litra Finance. **During the audit, we assume the user could be malicious, which means all messages sent to Litra Finance are untrusted.**

**We enumerated the attack surface based on this assumption.**

## 1.5 About Us

Supremacy is a leading blockchain security firm, composed of industry hackers and academic researchers, provide top-notch security solutions through our technology precipitation and innovative research.

We are reachable at X (<https://x.com/SupremacyHQ>), or Email ([contact@supremacy.email](mailto:contact@supremacy.email)).

## 1.6 Terminology

For the purpose of this assessment, we adopt the following terminology. To classify the severity of our findings, we determine the likelihood and impact (according to the CVSS risk rating methodology).

- Likelihood represents the likelihood of a finding to be triggered or exploited in practice
- Impact specifies the technical and business-related consequences of a finding
- Severity is derived based on the likelihood and the impact

We categorize the findings into four distinct categories, depending on their severity. These severities are derived from the likelihood and the impact using the following table, following a standard risk assessment procedure.

		Severity		
Impact	High	Critical	High	Medium
	Medium	High	Medium	Low
	Low	Medium	Low	Low
		High	Medium	Low
		Likelihood		

As seen in the table above, findings that have both a high likelihood and a high impact are classified as critical. Intuitively, such findings are likely to be triggered and cause significant disruption. Overall, the severity correlates with the associated risk. However, every finding's risk should always be closely checked, regardless of severity.

## 2 Findings

The table below summarizes the findings of the audit, including status and severity details.

ID	Severity	Description	Status
1	Critical	WrappedNFT forcible minting	Fixed
2	Critical	Theft of any user's NFT	Fixed
3	Medium	Potential fee loss	Fixed
4	Medium	Centralization risk	Confirmed
5	Low	Unchecked zero-address	Fixed
6	Low	Unchecked return values	Fixed
7	Informational	Missing event records	Fixed
8	Informational	Best Practices	Fixed
9	Informational	Best Practices	Fixed

## 2.1 Critical

### 1. WrappedNFT forcible minting [Medium]

Severity: Critical

Likelihood: High

Impact: High

Status: Fixed

#### Description

NFTVault::wrap() is a function that converts ERC721 (NFT) to ERC20 (WrappedNFT). However, since #L90 assigns wnftId to wnftIds[WrappedNFT] and \_nftAddr is controllable, resulting in WrappedNFT being arbitrarily mint.

1. The hacker selects an already existing WrappedNFT series and calls NFTVault::wrap(WrappedNFT, 0).
2. Then #L56 of wrap() will call WrappedNFT::transferFrom() externally, and since WrappedNFT is a standard ERC20 Token, it can be executed normally, but without actually transferring the Token, because \_value is 0.
3. Since WrappedNFT can obtain wnftId through #L61, it will not enter the CREATE procedure, but the else condition, and the wnft obtained through #L94 is WrappedNFT itself, so the subsequent procedure, will be directly for the caller Mint WrappedNFT.

```
47      /**
48       * @notice Wrap a NFT(ERC721) into a ERC20 token.
49       * @param _nftAddr address of NFT contract
50       * @param _tokenId token id of the NFT
51       */
52      function wrap (
53          address _nftAddr,
54          uint256 _tokenId
55      ) external payable nonReentrant {
56          IERC721(_nftAddr).transferFrom(msg.sender, address(this), _tokenId);
57          // Save nft record
58          uint256 recordId = wrappedNfts.length;
59          wrappedNfts.push(WrappedNFTInfo(_nftAddr, _tokenId, true));
60          // Get FT Info
61          uint256 wnftId = wnftIds[_nftAddr];
62          address wnft;
63          if(wnftId == 0) {
64              // Create a new FT
65              string memory wnftName;
66              string memory wnftSymbol;
67              (bool succeed, bytes memory result) =
68              _nftAddr.call(abi.encodeWithSignature("name()"));
69              if(succeed) {
70                  string memory nftName = abi.decode(result, (string));
71                  wnftName = string(abi.encodePacked(nftName, " Wrapped NFT"));
72              } else {
73                  wnftName = string(abi.encodePacked("Litra FT#", wnftId));
74              }
75              (succeed, result) =
76              _nftAddr.call(abi.encodeWithSignature("symbol()"));
77              if(succeed) {
78                  string memory nftSymbol = abi.decode(result, (string));
79                  wnftSymbol = string(abi.encodePacked(nftSymbol, "wnft"));
80              } else {
81                  wnftSymbol = string(abi.encodePacked("LWNFT#", wnftId));
```



```

80         }
81         wnft = address(new WrappedNFT(wnftName, wnftSymbol));
82         // get ftId
83         uint256 _nextWnftId = nextWnftId;
84         wnftId = _nextWnftId;
85         _nextWnftId ++;
86         nextWnftId = _nextWnftId;
87         // storage
88         wnfts[wnftId] = WNFTInfo(_nftAddr, wnft);
89         wnftIds[_nftAddr] = wnftId;
90         wnftIds[wnft] = wnftId;
91
92         emit CreateWrappedNFT(_nftAddr, wnftId, wnft);
93     } else {
94         wnft = wnfts[wnftId].wnftAddr;
95     }
96     // bound FT and NFT
97     _nfts[wnftId].add(recordId);
98     // mint and charge fee
99     uint256 fee;
100    if(address(feeManager) != address(0)) {
101        fee = feeManager.wrapFee(wnft);
102    }
103    if(fee > 0) {
104        WrappedNFT(wnft).mint(address(feeManager), fee);
105    }
106    WrappedNFT(wnft).mint(msg.sender, 1e18 - fee);
107
108    emit Wrap(msg.sender, wnftId, recordId);
109 }

```

## NFTVault.sol

### Recommendation

Delete #L90 from NFTVault.sol.

### 2. Theft of any user's NFT [Critical]

Severity: Critical

Likelihood: High

Impact: High

Status: Fixed

### Description

Based on the premise of **Critical-1**, a hacker can force the minting of any number of WrappedNFTs under a certain NFT series. However, in the NFTVault::unwrap() function, it allows the user to submit 1e18 WrappedNFTs and redeem the NFTs of that series. thus, the hacker can premeditatedly obtain all the deposited NFTVault in NFTVault and thus theft all users' NFTs by calling NFTVault::unwrap().

```

128     /**
129         @notice Redeem nft from vault and burn one FT
130         @param _wnftId index of fts
131         @param _nftId Greater than or equal 0 to redeem a designated nft with a
132             more fees
133             Less than 0 to redeem a recent fungiblized nft with a
134             normal fee
135     */

```

```

134     function unwrap(uint256 _wnftId, uint256 _nftId) external payable
nonReentrant {
135         WNFTInfo memory ftInfo = wnfts[_wnftId];
136         require(ftInfo.nftAddr != address(0), "Invalid FT");
137         require(WrappedNFT(ftInfo.wnftAddr).balanceOf(msg.sender) >= 1e18,
"Insufficient ft");
138         require(_nfts[_wnftId].length() > 0, "No NFT in vault");
139         require(_nfts[_wnftId].contains(uint256(_nftId)), "Invalid nftId");
140         // burn ft and charge fee
141         uint256 fee;
142         if(address(feeManager) != address(0)) {
143             fee = feeManager.unwrapFee(ftInfo.wnftAddr);
144         }
145         if(fee > 0) {
146             WrappedNFT(ftInfo.wnftAddr).transferFrom(msg.sender,
address(feeManager), fee);
147         }
148         WrappedNFT(ftInfo.wnftAddr).transferFrom(msg.sender, address(this),
1e18);
149         WrappedNFT(ftInfo.wnftAddr).burn(1e18);
150         // return nft
151         WrappedNFTInfo memory nftInfo = wrappedNfts[_nftId];
152         wrappedNfts[_nftId].inVault = false;
153         _nfts[_wnftId].remove(_nftId);
154         IERC721(nftInfo.nftAddr).safeTransferFrom(address(this), msg.sender,
nftInfo.tokenId);
155
156         emit Unwrap(msg.sender, _wnftId, _nftId);
157     }

```

## NFTVault.sol

### Recommendation

Refer to the **Critical-1** recommendation.

## 2.2 Medium

### 3. Potential fee loss [Medium]

Severity: Medium

Likelihood: High

Impact: Low

Status: Fixed

### Description

It is not recommended to leave the initial fee to the user to set, because due to the atomic nature of the transaction, it is not possible to charge a fee for WrappedNFTs created within a single transaction anyway. If the user sets both Wrap and Unwrap fees to 0 after the WrappedNFT is created, no fee will be charged during the window until the parameterAdmin sets the fee again, thus, causing some financial loss.

```

91     /**
92         @notice Set fee for wrapping.
93         Anyone can make the first setting, but generally the first maker will be
creator of wnft.
94         After first setting, only parameter admin can change

```

```

95     */
96     function setWrapFee(address _wnft, uint256 _fee) external {
97         Fee memory fee = _wrapFee[_wnft];
98         require(!fee.initialized || msg.sender == parameterAdmin, "! parameter
admin");
99         _wrapFee[_wnft] = Fee(true, _fee);
100     }
101
102     /**
103         @notice Set fee for unwrapping.
104         Anyone can make the first setting, but generally the first maker will be
creator of wnft.
105         After first setting, only parameter admin can change
106     */
107     function setUnwrapFee(address _wnft, uint256 _fee) external {
108         Fee memory fee = _unwrapFee[_wnft];
109         require(!fee.initialized || msg.sender == parameterAdmin, "! parameter
admin");
110         _unwrapFee[_wnft] = Fee(true, _fee);
111     }

```

### FeeManager.sol

## Recommendation

When the user calls NFTVault::Wrap() OR NFTVault::Unwrap(), call FeeManager::setWrapFee() externally to set the fee instantly before NFTVault transfers the fee (need to add access control in FeeManager).

## 4. Centralization risk [Medium]

Severity: Medium

Likelihood: Low

Impact: High

Status: Confirmed

## Description

In the Litra Finance, privileged accounts exist that play a key role in managing and regulating the operation of the entire system (e.g., configuring various parameters and setting stopped parameters). It also has the privilege of controlling or managing the flow of assets managed by the protocol.

Our analysis shows that privileged accounts need to be scrutinized. In the following, we will examine privileged accounts and the associated privileged access in the current contract.

Note that if the privileged owner account is a plain EOA, this may be worrisome and pose counter-party risk to the protocol users. A multi-sig account could greatly alleviate this concern, though it is still far from perfect. Specifically, a better approach is to eliminate the administration key concern by transferring the role to a community-governed DAO. In the meantime, a timelock-based mechanism can also be considered as mitigation.

```

3 contract OwnershipAdminManaged {
4     address public ownershipAdmin;

```

```

5     address public futureOwnershipAdmin;
6
7     constructor(address _o) {
8         ownershipAdmin = _o;
9     }
10
11    modifier onlyOwnershipAdmin {
12        require(msg.sender == ownershipAdmin, "! ownership admin");
13        _;
14    }
15
16    function commitOwnershipAdmin(address _o) external onlyOwnershipAdmin {
17        futureOwnershipAdmin = _o;
18    }
19
20    function applyOwnershipAdmin() external {
21        require(msg.sender == futureOwnershipAdmin, "Access denied!");
22        ownershipAdmin = futureOwnershipAdmin;
23    }
24 }

```

### OwnershipAdminManaged.sol

```

1  pragma solidity ^0.8.0;
2
3  import "./OwnershipAdminManaged.sol";
4
5  abstract contract EmergencyAdminManaged is OwnershipAdminManaged {
6      address public emergencyAdmin;
7      address public futureEmergencyAdmin;
8
9      constructor(address _e) {
10         emergencyAdmin = _e;
11     }
12
13    modifier onlyEmergencyAdmin {
14        require(msg.sender == emergencyAdmin, "! emergency admin");
15        _;
16    }
17
18    function commitEmergencyAdmin(address _e) external onlyEmergencyAdmin {
19        futureEmergencyAdmin = _e;
20    }
21
22    function applyEmergencyAdmin() external {
23        require(msg.sender == futureEmergencyAdmin, "! emergency admin");
24        emergencyAdmin = futureEmergencyAdmin;
25    }
26 }

```

### EmergencyAdminManaged.sol

```

1  pragma solidity ^0.8.0;
2
3  import "./OwnershipAdminManaged.sol";

```

```

4
5 abstract contract ParameterAdminManaged is OwnershipAdminManaged {
6     address public parameterAdmin;
7     address public futureParameterAdmin;
8
9     constructor(address _e) {
10         parameterAdmin = _e;
11     }
12
13     modifier onlyParameterAdmin {
14         require(msg.sender == parameterAdmin, "! parameter admin");
15         _;
16     }
17
18     function commitParameterAdmin(address _p) external onlyOwnershipAdmin {
19         futureParameterAdmin = _p;
20     }
21
22     function applyParameterAdmin() external {
23         require(msg.sender == futureParameterAdmin, "Access denied!");
24         parameterAdmin = futureParameterAdmin;
25     }
26 }

```

ParameterAdminManaged.sol

## Recommendation

Initially onboarding could use multisign wallets or timelocks to initially mitigate centralization risks, but as a long-running protocol, we recommend eventually transfer the privileged account to the intended DAO-like governance contract. All privileged operations may need to be mediated with necessary timelocks.

Eventually, activate the normal on-chain community-based governance life-cycle and ensure the intended trustless nature and high-quality distributed governance.

## 2.3 Low

### 5. Unchecked zero-address [Low]

Severity: Low

Likelihood: Low

Impact: Low

Status: Fixed

## Description

Adding the requirement for non-zero address to target.

EmergencyAdminManaged::commitEmergencyAdmin()

OwnershipAdminManaged::commitOwnershipAdmin()

ParameterAdminManaged::commitParameterAdmin()

## Recommendation

Consider adding zero address validation and non-previous address validation.

## 6. Unchecked return values [Low]

Severity: Low

Likelihood: Low

Impact: Low

Status: Undetermined

### Description

The NFTVault contract uses the EnumerableSet library to add and remove values to the set. Both functions return boolean values when they're called.

```
47      /**
48         @notice Wrap a NFT(IERC721) into a ERC20 token.
49         @param _nftAddr address of NFT contract
50         @param _tokenId token id of the NFT
51      */
52      function wrap (
53          address _nftAddr,
54          uint256 _tokenId
55      ) external payable nonReentrant {
56          IERC721(_nftAddr).transferFrom(msg.sender, address(this), _tokenId);
57          // Save nft record
58          uint256 recordId = wrappedNfts.length;
59          wrappedNfts.push(WrappedNFTInfo(_nftAddr, _tokenId, true));
60          // Get FT Info
61          uint256 wnftId = wnftIds[_nftAddr];
62          address wnft;
63          if(wnftId == 0) {
64              // Create a new FT
65              string memory wnftName;
66              string memory wnftSymbol;
67              (bool succeed, bytes memory result) =
68              _nftAddr.call(abi.encodeWithSignature("name()"));
69              if(succeed) {
70                  string memory nftName = abi.decode(result, (string));
71                  wnftName = string(abi.encodePacked(nftName, " Wrapped NFT"));
72              } else {
73                  wnftName = string(abi.encodePacked("Litra FT#", wnftId));
74              }
75              (succeed, result) =
76              _nftAddr.call(abi.encodeWithSignature("symbol()"));
77              if(succeed) {
78                  string memory nftSymbol = abi.decode(result, (string));
79                  wnftSymbol = string(abi.encodePacked(nftSymbol, "wnft"));
80              } else {
81                  wnftSymbol = string(abi.encodePacked("LWNFT#", wnftId));
82              }
83              wnft = address(new WrappedNFT(wnftName, wnftSymbol));
84              // get ftId
85              uint256 _nextWnftId = nextWnftId;
86              wnftId = _nextWnftId;
87              _nextWnftId ++;
88              nextWnftId = _nextWnftId;
89              // storage
90              wnfts[wnftId] = WNFTInfo(_nftAddr, wnft);
91              wnftIds[_nftAddr] = wnftId;
92              wnftIds[wnft] = wnftId;
93              emit CreateWrappedNFT(_nftAddr, wnftId, wnft);
94          } else {
95              wnft = wnfts[wnftId].wnftAddr;
96          }
97          // bound FT and NFT
98          _nfts[wnftId].add(recordId);
```

```

98         // mint and charge fee
99         uint256 fee;
100         if(address(feeManager) != address(0)) {
101             fee = feeManager.wrapFee(wnft);
102         }
103         if(fee > 0) {
104             WrappedNFT(wnft).mint(address(feeManager), fee);
105         }
106         WrappedNFT(wnft).mint(msg.sender, 1e18 - fee);
107
108         emit Wrap(msg.sender, wnftId, recordId);
109     }

```

## NFTVault.sol

```

128     /**
129         @notice Redeem nft from vault and burn one FT
130         @param _wnftId index of fts
131         @param _nftId Greater than or equal 0 to redeem a designated nft with a
132         more fees
133         Less than 0 to redeem a recent fungiblized nft with a
134         normal fee
135     */
136     function unwrap(uint256 _wnftId, uint256 _nftId) external payable
137     nonReentrant {
138         WNFTInfo memory ftInfo = wnfts[_wnftId];
139         require(ftInfo.nftAddr != address(0), "Invalid FT");
140         require(WrappedNFT(ftInfo.wnftAddr).balanceOf(msg.sender) >= 1e18,
141             "Insufficient ft");
142         require(_nfts[_wnftId].length() > 0, "No NFT in vault");
143         require(_nfts[_wnftId].contains(uint256(_nftId)), "Invalid nftId");
144         // burn ft and charge fee
145         uint256 fee;
146         if(address(feeManager) != address(0)) {
147             fee = feeManager.unwrapFee(ftInfo.wnftAddr);
148         }
149         if(fee > 0) {
150             WrappedNFT(ftInfo.wnftAddr).transferFrom(msg.sender,
151                 address(feeManager), fee);
152         }
153         WrappedNFT(ftInfo.wnftAddr).transferFrom(msg.sender, address(this),
154             1e18);
155         WrappedNFT(ftInfo.wnftAddr).burn(1e18);
156         // return nft
157         WrappedNFTInfo memory nftInfo = wrappedNfts[_nftId];
158         wrappedNfts[_nftId].inVault = false;
159         _nfts[_wnftId].remove(_nftId);
160         IERC721(nftInfo.nftAddr).safeTransferFrom(address(this), msg.sender,
161             nftInfo.tokenId);
162
163         emit Unwrap(msg.sender, _wnftId, _nftId);
164     }

```

Therefore, there should be a check to validate that the addition or removal from the set was correct. Otherwise, every time the function is called, the owner will have to check using getter functions.

## NFTVault.sol

## Recommendation

Revise the code logic accordingly.

## 2.4 Informational

### 7. Missing event records [Informational]

Status: Fixed

#### Description

In the EmergencyAdminManaged, OwnershipAdminManaged, ParameterAdminManaged, and Stoppable contracts, privileged accounts can set Admin privileges and stopped status through privileged functions respectively, but no event logging is performed. And in FeeManager contract, user will call setWrapFee() & setUnwrapFee() to change the status, but no event logging.

```
47 pragma solidity ^0.8.0;
48
49 import "./OwnershipAdminManaged.sol";
50
51 abstract contract EmergencyAdminManaged is OwnershipAdminManaged {
52     address public emergencyAdmin;
53     address public futureEmergencyAdmin;
54
55     constructor(address _e) {
56         emergencyAdmin = _e;
57     }
58
59     modifier onlyEmergencyAdmin {
60         require(msg.sender == emergencyAdmin, "! emergency admin");
61         _;
62     }
63
64     function commitEmergencyAdmin(address _e) external onlyEmergencyAdmin {
65         futureEmergencyAdmin = _e;
66     }
67
68     function applyEmergencyAdmin() external {
69         require(msg.sender == futureEmergencyAdmin, "! emergency admin");
70         emergencyAdmin = futureEmergencyAdmin;
71     }
72 }
73
```

EmergencyAdminManaged.sol

Events are important because off-chain monitoring tools rely on them to index important state changes to the smart contract(s).

## Recommendation

Consider emitting events when state changes are performed in the EmergencyAdminManaged, OwnershipAdminManaged, ParameterAdminManaged, Stoppable and FeeManager contract.



## 8. Best Practices [Informational]

Status: Fixed

### Description

Some storage variables should be immutable. Marking these as immutable (as they never change outside the constructor) would avoid them taking space in the storage.

```
1 // SPDX-License-Identifier: MIT
2 pragma solidity ^0.8.0;
3
4 import "../admin/Stoppable.sol";
5 import "../interfaces/IBurner.sol";
6 import "../interfaces/IFeeManager.sol";
7 import "../admin/ParameterAdminManaged.sol";
8
9 import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
10
11 contract FeeManager is IFeeManager, Stoppable, ParameterAdminManaged {
12     struct Fee {
13         bool initialized;
14         uint256 value;
15     }
16
17     address public vault;
18     ...
19 }
```

FeeManager.sol

### Recommendation

Revise the code logic accordingly.

## 9. Best Practices [Informational]

Status: Fixed

### Description

In the EmergencyAdminManaged, OwnershipAdminManaged and ParameterAdminManaged contracts, privileged accounts can each set administrative privileges via privileged functions, where a transition account future exists so that the future account in the future can be upgraded to administrator by calling the applyAdmin() privilege function.

However, the call to applyAdmin() does not reset future, making the calling account both Admin and future.

```
1 pragma solidity ^0.8.0;
2
3 import "../OwnershipAdminManaged.sol";
4
5 abstract contract EmergencyAdminManaged is OwnershipAdminManaged {
6     address public emergencyAdmin;
7     address public futureEmergencyAdmin;
```

```

8
9     constructor(address _e) {
10         emergencyAdmin = _e;
11     }
12
13     modifier onlyEmergencyAdmin {
14         require(msg.sender == emergencyAdmin, "! emergency admin");
15         _;
16     }
17
18     function commitEmergencyAdmin(address _e) external onlyEmergencyAdmin {
19         futureEmergencyAdmin = _e;
20     }
21
22     function applyEmergencyAdmin() external {
23         require(msg.sender == futureEmergencyAdmin, "! emergency admin");
24         emergencyAdmin = futureEmergencyAdmin;
25     }
26 }

```

EmergencyAdminManaged.sol

### Recommendation

Revise the code logic accordingly.

### 3 Disclaimer

This security audit report does not constitute investment advice or a personal recommendation. It does not consider, and should not be interpreted as considering or having any bearing on, the potential economics of a token, token sale or any other product, service or other asset. Any entity should not rely on this report in any way, including for the purpose of making any decisions to buy or sell any token, product, service or other asset. This security audit report is not an endorsement of any particular project or team, and the report does not guarantee the security of any particular project. This audit does not give any warranties on discovering all security issues of the smart contracts, i.e., the evaluation result does not guarantee the nonexistence of any further findings of security issues, also cannot make guarantees about any additional code added to the assessed project after the audit version. As one audit-based assessment cannot be considered comprehensive, we always recommend proceeding with independent audits and a public bug bounty program to ensure the security of smart contract(s). Unless explicitly specified, the security of the language itself (e.g., the solidity language), the underlying compiling toolchain and the computing infrastructure are out of the scope.